
第1章 数の計算

1.1 Maximaによる計算
Maximaを起動すると，大きな窓になにやら英語が表示されます．1

まず最初に，

1+2

と打ち込み，(Shift)を押しながら (Enter)を入力してみたください．画面に� �
(%i1) 1+2;

(%o1) 3� �
と表示されたと思います．

最初の行は入力が 1+2であったことを表し，2番目の行はその出力結果で

ある 3を表示しています．

この様に，Maximaは命令を打ち込んだ後，(Shift)キーを押しながら (Enter)

を入力することにより，その結果を表示するようになっています．(Shift)キー

を押さずに (Enter)を入力すると，単に改行されるだけです．

画面をよく見ると，入力ラインは (%i1)のような%iの後に番号がついたラ

ベルが与えられます．一方，出力ラインには (%o1)のような%oの後に番号が

ついたラベルが与えられます．このラベルの iは inputの iで，oは output

の oを意味します．一般に n番目の入力には (%in)というラベルが振られ，

n番目の出力には (%on)というラベルが振られます．

1.2 加減乗除
加減乗除にはそれぞれ +, −, ∗, /を用います．出力は下のようになりま

す．Maximaは上の例のように電卓のような使い方ができます．例えば (3×
(24− 8) + 4)÷ 18は

(3*(24-8)+4)/18

のように入力します．出力は次のようになり，計算結果
26

9
が表示されます．

1以前のバージョンでは，命令を入力するための入力窓は出力を表示する出力窓と別になって
いましたが，現在のバージョンでは分かれていません

� �
(%i2) (3*(24-8)+4)/18;

(%o2)
26

9� �
さて，上の例では計算結果は分数で表示されました．これは計算式の中の

数がすべて整数のためです．結果をを小数で表示するには，この後に次のよ

うに入力します．� �
(%i3) %, numer;

(%o3) 2.888888888888889� �
入力行の%はすぐ手前の出力，ここでは (%o2)の

26

9
を表し，そのあとのコ

ンマと numerでその値の小数近似を表示します．たとえば，5/2, numerと

入力すると，次のような出力を得ます．� �
(%i4) 5/2, numer;

(%o4) 2.5� �
先ほどの計算 (3× (24− 8) + 4)÷ 18において，ひとつでも小数が含まれ

れば，出力は小数になります．たとえば，(%i2)の 3を 3.0に変えて計算を

実行すると次のようになります．� �
(%i5) (3.0*(24-8)+4)/18;

(%o5) 2.888888888888889� �

1.3 巾，階乗
巾，階乗などの計算は^, ! を用います．例えば 324, 10!は� �
(%i6) 3^24; 10!;

(%o6) 282429536481

(%o7) 3628800� �
2



のようになります．

上の計算では，324と 10!が１行で入力されています．このように，複数の

計算を１行で入力するときは，セミコロン ; を用いて，式を区切ってやりま

す．出力 (%o6)が 324 の計算結果で，(%o7)が 10!の計算結果になります．

この後に命令を入力すると (%i8)という番号がつき，(%i7)は一見スキッ

プされたように見えます．実は，(%i6)で二つの計算を実行していますが，実

際には (%i6) での入力は 324 のみで，10!は (%i7)の入力と考えます．その

ため，(%o7)が 10!の計算結果なのです．

1.4 平方根
平方根がついた数の計算もできます．(1 +

√
2)5を計算しましょう．

√
2は

sqrt(2)とします．� �
(%i8) (1 + sqrt(2))^5;

(%o8) (
√
2 + 1)5

(%i9) expand (%);

(%o9) 29
√
2 + 41

(%i10) %, numer;

(%o10) 82.01219330881976� �
入力行 (%i9)の expand(%)について説明します．%は前の出力 (

√
2 + 1)5

を表します．そして expandという命令は，その式を展開しなさいというも

のです．展開した結果は 29
√
2 + 41であることが分かります．さらに次の行

では，その値を小数で近似しています．ちなみに，窓の下に Expand と書か

れたボタンがありますが，これを押すことで自動的に expand(%)が入力され

ます．

入力する際に%を用いると，一つ前の出力結果を参照することができます

が，Maximaは一つ前ばかりでなく，行についているラベルを用いてその行

の出力結果を参照することができます．たとえば (%o4)の 2.5を 3倍してみ

ます．� �
(%i11) (%o4)*3;

(%o11) 7.5� �
一度入力した命令も参照できます．たとえば，(%i6)の 10!を再実行して

みます．ただし，この場合は，先ほど説明したように，(%i6)の 2番目の入

力なので，実際には (%i7)として処理します．

� �
(%i12) (%i7);

(%o12) 3628800� �
1.5 文字列への代入
Maximaでは任意の文字列に数値を代入することができます．たとえば文

字 xに 12を代入するときは x:12のようにコロン : を用います．� �
(%i13) x:12$ y1:2$ y2:3$

(%i16) x*y1; x*y2;

(%o16) 24

(%o17) 36

(%i18) kill(x,y1,y2);

(%o18) done

(%i19) x*y1; x*y2;

(%o19) xy1

(%o20) xy2� �
上の例では，(%i13)で x, y1, y2にそれぞれ 12, 2, 3を代入し，(%i16)

で，xの y1倍と y2倍を計算しています．その後，(%i18)の kill(x,y1,y2)

で x, y1, y2の値をクリアしています．(%i19)で同じ計算をしていますが，

(\o19), (\o20) の出力では文字式が表示されるだけになっています．

ちなみに，(%i13)で$が用いられています．通常は，これはセミコロン ;

が用いられるところですが，そのようにすると 12, 2, 3がその後に表示され，

多少うっとおしいです．セミコロンの代わりに$を用いると，出力結果が表示

されません．

1.6 商と余り
割り算では /以外に，商と余りを計算する方法があります．例えば 54÷5 =

10 · · · 4の計算で，商の 10と余りの 5は次のように計算します．� �
(%i21) divide(54,5);

(%o21) [10, 4]� �

(%o21)の [10,4]は最初の 10が商で 2番目の 4が余りを表します．これか

ら商と余りを取り出すには次のようにします．� �
(%i22) %[1];

(%o22) 10

(%i23) (%o21)[2];

(%o23) 4� �
一般に，[a,b,...,c]のような列の第k番目を取り出すには[a,b,...,c][k]

のようにします．上の例では，divide(54,5)[1], divide(54,5)[2]のよう

にしても同じ結果を得ます．

1.7 最大公約数，最小公倍数
与えられた 2整数 a, bに対して，それらの最大公約数は gcd(a,b)で求ま

ります．� �
(%i24) gcd(360,945);

(%o24) 45� �
今度は 3つの数 360,945,2310の最大公約数を求めてみましょう� �
(%i25) gcd(360,945,2310);

なにやら文字列� �
エラーになったようです．3つ以上の数の最大公約数は，gcd(a,b,c,...)

のようにしては求まりません．次のようにします．� �
(%i26) gcd(gcd(360,945),2310);

(%o26) 15� �
一方，最小公倍数は lcm(a,b)で求まりますが，この関数を用いるために

は，最初に functsというライブラリをロードしておかなくてはなりません．� �
(%i27) load("functs");

(%o27) なにやら文字列

(%i28) lcm(20,32,86);

(%o28) 6880� �

lcmは 3つ以上の数でも問題なかったようです．

1.8 Euclidの互助法
ここでは Euclidの互助法を用いて gcdを用いずに与えられた 2つの正の

整数の最大公約数を求めてみましょう．

補題 1.8.1. a, bは正の整数で，a ≥ bを満たすものとする．いま aは bの倍

数でないとし，aを bで割った余りが r (1 ≤ r < b)であるなら，aと bの最

大公約数と，bと rの最大公約数は一致する．

上の補題を用いて，与えられた正の整数 aと bの最大公約数が次のように

して求まります．

1. a ≥ bとし，aを bで割った余りを r1 とする．

2. もし r1 = 0であれば，aは bで割り切れることになり，最大公約数は b

になる．

3. r1 ̸= 0のとき，上の補題より aと bの最大公約数と bと r1の最大公約

数は一致するから，bと r1 の最大公約数を求めればよい．

4. 次に上の手順を，a, bの代わりに，b, r1を用いて行う．すなわち，bが

r1で割り切れれば r1が最大公約数になり，割り切れなければ，bを r1

で割った余り，それを r2とする，を考えると，r1と r2の最大公約数が

bと r1 の最大公約数，すなわち aと bの最大公約数と一致する．

5. これを繰り返していくと，a ≥ b > r1 > r2 > r3 > · · · > rj = 0 という

列が得られる．ただし，rtは rt−2を rt−1で割った余りである．そこで

rj−1 が最初の a, bの最大公約数になる．

1155と 1008の最大公約数を求めてみましょう．� �
(%i29) a:1155$ b:1008$

(%i30) r1:divide(a,b)[2]

(%o30) 147

(%i31) r2:divide(b,r1)[2]

(%o31) 126

(%i32) r3:divide(r1,r2)[2]

(%o32) 21

(%i33) r4:divide(r2,r3)[2]

(%o33) 0� �
よって最大公約数は，r3の 21になることがわかります．

演習問題
1. 次の値を簡単にせよ．また，それぞれを小数で近似せよ．

(1)
2 + 312 − 5!

7
(2)

(
1 +

√
3
)5

2. 変数 x, y, zに x = 3, y = 12, z = −5を代入し，次の値を求めよ．

(1) x2 − 3yz (2) (x− 2y)(5z + xy)

3. 上で代入した変数 x, y, zの値を削除し，新たに x =
√
3, y = 1.2, z = 7

を代入し，上の (1), (2)の値を求めよ．

4. gcd, lcmを用いずに 8775, 4590の最大公約数および最小公倍数を求め

よ．（ヒント：a, bの最大公約数を d，最小公倍数をmとすると，ab = dm

が成り立つ．）

5. 上の問題と同様に，gcd, lcmを用いず 8775, 4590, 1970の最大公約数

および最小公倍数を求めよ．

第2章 式の計算

2.1 文字式の計算
Maximaは文字式の計算が可能です．たとえば (x2−3x+1)−(2x2+4x−5)

の計算は次のようにします．� �
(%i1) (x^2-3*x+1)-(2*x^2+4*x-5);

(%o1) −x2 − 7x+ 6� �
次の例は，A = x2 + 3y2 − xy + 5x, B = −x+ 3y2 + 4xyと代入して計算

しています．� �
(%i2) A:x^2+3*y^2-x*y+5*x$ B:-x+3*y^2+4*x*y$

(%i4) 2*A-B;

(%o4) 2(3y2 − xy + x2 + 5x)− 3y2 − 4xy + x

(%i5) expand(%);

(%o5) 3y2 − 6xy + 2x2 + 11x� �
多項式を指定した文字で整理する場合は ratsimp を用いて次のようにし

ます．� �
(%i6) pol:x^2*y-3*x+5*y-x^2+1$ ratsimp(pol,x)

(%o7) 5y + x2(y − 1)− 3x+ 1

(%i8) ratsimp(pol,y)

(%o8) (x2 + 5)y − x2 − 3x+ 1� �
(%o7)は xについて整理しているのですが，多少違和感がある出力になっ

ています．このように，必ずしも期待する結果にならないことがあります．

多項式の変数に値を代入するにはいくつかの方法があります．polの xに

2を代入してみます．

� �
(%i9) pol,x=2

(%o9) 9y − 9

(%i10) ev(pol,x=2)

(%o10) 9y − 9

(%i11) subst(2,x,pol);

(%o11) 9y − 9� �
次の例では，最初に多項式 polの xに yを代入しています．次に xに a+b

を，yに a-bを代入し，その後に展開してから文字 bについて整理しています．� �
(%i12) pol,x=y;

(%o12) y3 − y2 + 2y + 1

(%i13) pol,[x=a+b, y=a-b];

(%o13) (a− b)(b+ a)2 − (b+ a)2 − 3(b+ a) + 5(a− b) + 1

(%i14) ratsimp(expand(%),b);

(%o14) −b3 + (−a− 1)b2 + (a2 − 2a− 8)b+ a3 − a2 + 2a+ 1� �

2.2 展開と因数分解
式の展開は expandを用いました．逆に与えられた式の因数分解をするに

は factorを用います．� �
(%i15) expand((x^3-3*x^2+2)*(4*x+5));

(%o15) 4x4 − 7x3 − 15x2 + 8x+ 10

(%i16) factor(%);

(%o16) (x− 1)(4x+ 5)(x2 − 2x− 2)

(%i17) (3*x^3+5*x^2-11*x+3)/(3*x-1);

(%o17)
3x3 + 5x2 − 11x+ 3

3x− 1

(%i18) factor(%);

(%o18) (x− 1)(x+ 3)� �
最後の例にあるように，分数に factorを適応すると，因数分解をして約

分してくれます．

演習問題
1. A = x2 − 2x+ 3, B = −3x2 + 1, C = x2 − 2x+ 5のとき，次の式を計

算せよ．

(1) (B −A) + (C −A) (2) 2C − (B + C − 2A)

2. 次の式を展開せよ．

(1) (x− y)(x+ y)(x2+ y2) (2) (x2 − 1)(x2 + x+ 1)(x2 − x+ 1)

3. 次の式を因数分解せよ．

(1) 3x2y − 7xy − 6y (2) (x− y)(x− y + 5) + 6

4. x = 2 +
√
3, y = 2−

√
3のとき，次の式の値を求めよ．

(1) x2 + y2 (2)
y2

x
+

x2

y

5. a+ b+ c = 0, abc ̸= 0のとき，次の等式を証明せよ．

(1)
b+ c

a
+

c+ a

b
+

a+ b

c
= −3

(2)
b2 − c2

a
+

c2 − a2

b
+

a2 − b2

c
= 0

（ヒント：与えられた条件を用いて左辺を変形すると右辺になればよい．）

第3章 方程式

3.1 代数方程式の解法
Maximaを用いて方程式を解くにはsolveを用います．たとえばx2+x−5 =

0を解くのには次のようにします．� �
(%i1) solve(x^2+x-5=0);

(%o1) [x = −
√
21 + 1

2
, x =

√
21− 1

2
]

(%i2) ans:%$ ans[2];

(%o3) x =

√
21− 1

2

(%i4) x^2+x-5,%;

(%o4)
(
√
21− 1)2

4
+

√
21− 1

2
− 5

(%i5) expand(%);

(%o5) 0� �
最初の出力は，x2 +x− 5 = 0の解が−

√
21 + 1

2
と

√
21− 1

2
の 2つである

ことを意味しています．

2番目の入力の ans:% は，前の出力を ansという文字に代入しているとい

うことです．つまり ans =

[
x = −

√
21 + 1

2
, x =

√
21− 1

2

]
という意味です．

そしてその後の ans[2]で，ansの 2番目，すなわち x =

√
21− 1

2
を取り出

しています．

(%i4)は ans[2]である

√
21− 1

2
が実際に方程式の解になっているかどう

かを，元の式の xにその値を代入して調べています．そして，最後に上の式

を展開し，結果の 0が得られています．つまり

√
21− 1

2
が実際に解になって

いることがわかります．

こんどは x4 = 1を解きます．

� �
(%i6) solve(x^4=1);

(%o6) [ x = %i, x = −1, x = −%i, x = 1 ]� �
上の出力で%iは虚数単位

√
−1を表しています．つまり，方程式には複素

数の範囲で 4つの解 [i,−1,−i, 1]があることがわかります．

方程式を実数の範囲で解くには realrootsを用います．� �
(%i7) realroots(x^4=1);

(%o7) [ x = −1, x = 1 ]� �
連立方程式も solveを用いて解くことができます．たとえば未知数を xと

yとする連立方程式 {
x2 + y2 = 1

x+ y = 1

は，次のようにして解きます．� �
(%i8) solve([x^2+y^2=1, x+y=1]);

(%o8) [ [ y = 1, x = 0] , [ y = 0, x = 1] ]

(%i9) a:%$ a[1];

(%o10) [ y = 1, x = 0 ]

(%i11) a[1][2];

(%o11) x = 0� �
最初の出力は，解が y = 1, x = 0と y = 0, x = 1の 2組である事を意味し

ます．次に文字 aに最初の出力（解の集合）を代入し，a[1]で 1番目の解を

表示させています．最後の a[1][2]は，1番目の解の 2番目，つまり x = 0

を意味します．

演習問題
1. solveを用いて，2次方程式の解の公式を求めよ．同様に，n = 3, 4, 5

について，n次方程式の解の公式が求まるか試してみよ．

第4章 数列と級数

4.1 数列と数列の和
Maximaでは数列はリストとして与えます．たとえば2, 3, 5, 7, 11, 13, 17, 19, 23, 29

のような数列 aは次のようにして与えます．� �
(%i1) a: [2,3,5,7,11,13,17,19,23,29];

(%o1) [ 2, 3, 5, 7, 11, 13, 17, 19, 23, 29 ]� �
aの第 4項と第 9項を表示し，さらに初項から第 10項までの和を求めます．� �
(%i2) a[4]; a[9];

(%o2) 7

(%o3) 23

(%i4) sum(a[i],i,1,10);

(%o4) 129

(%i5) kill(a)$� �
最後のkill(a)で，aに代入した値（ここでは[2,3,5,7,11,13,17,19,23,29]）

を削除しています．

数列の第 n項を nの関数として定義できる場合もあります．関数の一般的

な話は後で説明しますが，たとえば，第 n項が 2n − 5である数列は，関数

b(x) = 2x− 5の xに自然数を代入したものと考えることができます．� �
(%i6) b(n):=2*n-5;

(%o6) b(n) := 2n− 5

(%i7) makelist(b(n),n,1,5);

(%o7) [−3,−1, 1, 3, 5 ]

(%i8) sum(b(i),i,1,n);

(%o8)

n∑
i=1

2i− 5

(%i9) %,simpsum;

(%o9) n2 − 4n� �

(%i7)の makelistはリストを作るコマンドです．初項から第５項までの

リストが表示されました．

また，初項から第 n項までの和
∑n

i=1 b(i)を計算していますが，(%i9)で

それを単純な形で表示しています．

この simpsumというのは，maximaに組み込まれている変数の一つで，通

常は falseという値が入っています．これは，次のようにすると確かめるこ

とができます．� �
(%i10) simpsum;

(%o10) false� �
(%i9)は，simpsumの値を一時的に true にして，和を表示させているこ

とになります．

常に，単純化した形で和を求めたいときは，次のようにします．� �
(%i11) simpsum:true;

(%o11) true

(%i12) sum(b(i),i,1,n);

(%o12) n2 − 4n� �
4.2 無限級数
無限級数の和も求めることができます．たとえば

∞∑
i=1

1

i2
は

� �
(%i13) sum(1/i^2, i,1,inf);

(%o13)
%pi2

6

(%i14) %, numer:

(%o14) 1.644934066848226� �
ここで，infは∞を，%piは円周率 πを表します．また，すでに simpsum

の値を trueにしてあるので，和は単純な形で表されています．

4.3 数列の極限
今度は，極限 lim

n→∞
(1 + 1/n)n を計算してみましょう．

� �
(%i15) limit((1+1/n)^n,n,inf);

(%o15) %e

(%i16) %, numer;

(%o16) %e

(%i17) float(%);

(%o17) 2.718281828459045� �
ここで%eはネピアの数（自然対数の底）です．(%i16)で小数近似を試み

ましたが失敗したようです．それで，(%i17)で floatという関数を用いて

みたら，今度はうまくいきました．

次に，極限 lim
n→∞

anを計算します．この極限は aの値により違ってきます．

すなわち

lim
n→∞

an =



∞ a > 1

1 a = 1

0 −1 < a < 1

不定 a = −1

複素無限大 a < −1

となります．最後の複素無限大の意味は，複素数における無限大で，細かい

話は避けますが，ここでは単に絶対値が無限大に発散すると考えていいと思

います．

最初に a > 1の仮定の下で計算してみます．仮定には assumeを用います．� �
(%i18) assume(a>1);

(%o18) [ a > 1]

(%i19) limit(a^n,n,inf);

(%o19) ∞� �
(%i18)では a > 1という仮定を aに与えています．

次に a = 1のときは自明なので，−1 < a < 1で計算します．そのために

は，先ほどの仮定 a > 1を削除する必要があります．仮定の削除は forget

を用います．

� �
(%i20) forget(a>1)

(%o20) [ a > 1]

(%i21) facts(a);

(%o21) [ ]

(%i22) assume(a<1,a>-1);

(%o22) [ a < 1, a > −1]

(%i23) limit(a^n,n,inf);

Is |a| − 1 positive, negative, or zero? n;

Is a positive, negative, or zero? p;

(%o23) 0� �
facts(a)は，現在 aに与えている仮定を表示させるものです．前の行で

forgetをして，仮定を削除してしまっているので，ここではブランクになっ

ています．

(%i23)の後に「Is |a| − 1 positive, negative, or zero?」という質問が表示

されます．これは |a| − 1 が正 (positive) なのか負 (negative) なのか，それ

とも 0 (zero) なのかという問い合わせです．−1 < a < 1を仮定しているわ

けですから，当然負に決まっているんですが，Maximaはその辺がちょっと

お馬鹿なようです．そこで，負になるので「n」と答えます．正であれば「p」

で，0であれば当然「0」です．次に aについても聞いてきますが，ここでは

とりあえず正の場合を考え「p」と答えました．ちなみに，aが負の場合でも

結果は同じになります．ここでは省略します．

最後に a = −1の場合と a < −1の場合を考えて見ます．� �
(%i24) limit((-1)^n,n,inf);

(%o24) ind

(%i25) forget(a<1,a>-1)

(%o25) [ a < 1, a > −1 ]

(%i26) assume(a<-1);

(%o26) [ a < −1 ]

(%i27) limit(a^n,n,inf);

(%o27) infinity� �
(%o24)の indは不定を表しています．そして，最後の (%o27)の infinity

が複素無限大で，通常の正の実無限大 inf と区別して下さい．

演習問題
1. 次の和を求めよ．（結果は因数分解すること）

(1)
n∑

i=1

i (2)
n∑

i=1

i2 (3)
n∑

i=1

i10 (4)
n∑

i=1

2i− 3i2 + 4i3

2. 初項から第 n項までの和が −n2 + 24nで与えられる数列の一般項を求

めよ．

3. すべての項が 1である数列を (a1n)n=1,2,... とおく．（つまり，任意の n

に対して a1n = 1）次に数列 {akn}, (k ≥ 1) を帰納的に

a2n =
n∑

i=1

a1i, a3n =
n∑

i=1

a2i, a4n =
n∑

i=1

a3i, . . .

のように定めたとき，{akn}の一般項を k = 1, 2, 3, 4, 5, 6について求め

よ．また，その結果から，一般に aknがどのようになるかを推測せよ．

ヒント：aknはMaximaではa[k](n)とする．すなわち，a1nはa[1](n):=1

と定める．

4. 第 n項が次で与えられる数列の極限を求めよ．

(1)
12 + 32 + · · ·+ (2n− 1)2

n3
(2) n1/n (3)

(
1− 1

n

)−n

5. a > 0のとき次の極限値を求めよ．

(1) lim
n→∞

an (2) lim
n→∞

a1/n

6. 初項が 1で正の公比を持つ等比数列 (fn)で漸化式 fn+2 = fn + fn+1

(n ≥ 1)を満たすものを求めよ．

7. 次の等式が成立することを確かめよ．(
1− 1

22

)(
1− 1

32

)
· · ·
(
1− 1

(n+ 1)2

)
=

n+ 2

2(n+ 1)

ヒント：数学的帰納法を用いて確かめる．

第5章 関数

5.1 関数の定義
関数の定義はすでに数列のところで説明しましたが，実は 2通りの定義の

方法があります．例えば f(x) = x2で定義される関数 f は次のように与える

ことができます．� �
(%i1) f(x):=x^2;

(%o1) f(x) := x2

(%i2) define(f(x),x^2);

(%o2) f(x) := x2� �
ただし，この 2つは完全に同じというわけではなく，微妙な違いがるため，

状況によって使い分ける必要があります．特に問題がなければ，前者の方が

簡単だし，記憶しやすいと思います．

上の関数の xにさまざまな値を代入してみます．� �
(%i3) f(-1); f(0); f(1/2); f(a+b); f(f(x))

(%o3) 1

(%o4) 0

(%o5)
1

4
(%o6) (b+ a)2

(%o7) x4� �
最後の f(f(x))は合成関数 f ◦ f(x)をあらわす事になります．

5.2 組み込み関数
Maximaには，三角関数や指数関数のような基本的な関数が組み込まれて

います．三角関数およびその逆関数については次のようなものが組み込まれ

ています．

sinx sin(x) cosx cos(x) tanx tan(x)

cscx csc(x) secx sec(x) cotx cot(x)

sin−1 x asin(x) cos−1 x acos(x) tan−1 x atan(x)

csc−1 x acsc(x) sec−1 x asec(x) cot−1 x acot(x)

また，指数関数 exは exp(x)または%e^xで，対数関数 log xは log(x)で表

されます．

sinπ, cos π
4 , tan

π
3 , arcsin 1, e

0, log 1の値を計算してみます．� �
(%i8) sin(%pi); cos(%pi/4); tan(%pi/3); asin(1);

(%o8) 0

(%o9)
1√
2

(%o10)
√
3

(%o11)
%pi

2
(%i12) exp(0); log(1);

(%o12) 1

(%o13) 0� �
加法定理を用いて，式に含まれる三角関数をできるだけやさしい形に展開

するには trigexpandを用います．その逆に，与えられた式を，ただ一つの

sinまたは cosのみを含む式の和として表すには trigreduceを用います．� �
(%i14) sin(x+y)*cos(x)^3;

(%o14) cos(x)3 sin(y + x)

(%i15) trigexpand(%);

(%o15) cos(x)3(cos(x) sin(y) + sin(x)) cos(y)

(%i16) trigreduce(%);

(%o16)
sin(y + 4x) + sin(y − 2x)

8
+

3 sin(y + 2x) + 3 sin(y)

8� �

5.3 関数の極限
関数の極限は，数列の極限と同様に求めることができます．たとえば

lim
x→0

sinx

x
, lim

x→∞
(
√
x2 + x+ 1− x)

は次のように計算します．

� �
(%i17) limit(sin(x)/x,x,0);

(%o17) 1

(%i18) limit(sqrt(x^2+x+1)-x, x, inf);

(%o18)
1

2� �
左極限や右極限も求めることができます．f(x) =

x− 1

|x− 1|
を考えて見ます．

この関数は

f(x) =

−1 x < 1

1 x > 1

であり，x = 1では定義されません．さらに，lim
x→1

f(x)は存在せず， lim
x→1+0

f(x) =

1, lim
x→1−0

f(x) = −1 になります．実際にMaximaで計算してみると次のよう

になります．� �
(%i19) f(x):=(x-1)/abs(x-1);

(%o19) f(x) :=
x− 1

|x− 1|
(%i20) limit(f(x),x,1);

(%o20) und

(%i21) limit(f(x),x,1,plus); limit(f(x),x,1,minus);

(%o21) 1

(%o22) −1� �
最初の行で，f(x)を定義しています．絶対値 |x − 1|は abs(x-1)としま

す．その後で，lim
x→1

f(x)を調べていますが，不定の undが出力されています．

さらに， lim
x→1+0

f(x)と lim
x→1−0

f(x) を調べたところ，きちんと 1と −1とい

う答えが返ってきました．

第6章 微分

6.1 常微分
微分には diffを用います．たとえば sinx, ex, log xの微分は次のように

します．� �
(%i1) diff(sin(x),x); diff(exp(x),x); diff(log(x),x);

(%o1) cos(x)

(%o2) %ex

(%o3)
1

x� �
自分で定義した関数も微分できます．� �
(%i4) f(x):=x^2+3*x-5$

(%i5) diff(f(x),x);

(%o5) 2x+ 3� �
以前に関数の定義が 2通りあり，微妙に違っていると書きました．その違

いは次のような場面であらわれてきます．� �
(%i6) g(x):=diff(f(x),x)$

(%i7) define(h(x),diff(f(x),x))$

(%i8) g(x);

(%o8) 2x+ 3

(%i9) h(x);

(%o9) 2x+ 3

(%i10) g(2);

エラーメッセージ

(%i11) h(2);

(%o11) 7� �
21



上のように g(x)も h(x)も一見同じであるように見えますが，xに値を代

入したときに違いが表れます．

関数の積や合成関数の微分もできます．� �
(%i12) diff(sin(x)*cos(x),x);

(%o12) cos(x)2 − sin(x)2

(%i13) diff(sin(log(x)),x);

(%o13)
cos(log(x))

x� �
高階の微分も計算できます．たとえば sin(x2)の 3階微分は次のようにし

ます．� �
(%i14) diff(sin(x^3),x,3);

(%o14) −54x3 sin(x3)− 27x6 cos(x3) + 6 cos(x3)� �
6.2 偏微分
偏微分もほとんど同様です．� �
(%i15) f(x,y,z):=sin(x)*cos(y)*z$

(%i16) diff(f(x,y,z),x); diff(f(x,y,z),y);

diff(f(x,y,z),z);

(%o16) cos(x) cos(y)z

(%o17) − sin(x) sin(y) ∗ z
(%o18) sin(x) cos(y)

(%i19) diff(diff(f(x,y,z),x),y);

(%o19) − cos(x) sin(y)z� �
最後は，xで偏微分した後 yで偏微分しています．つまり

∂2f(x, y, z)

∂y∂x
です．

演習問題
1. 関数 f(x) = xe−x2

(−5 ≤ x ≤ 5)の極値を求めよ．

第7章 積分

積分にはintegrateを用います．不定積分
∫

x2 sinx dx，定積分
∫ π/2

0

sinx dx

および広義積分
∫ ∞

0

e−x2

dxは次のように計算します．

� �
(%i1) integrate(x^2*sin(x)^2,x);

(%o1) − (6x2 − 3) sin(2x) + 6x cos(2x)− 4x3

24

(%i2) integrate(sin(x), x, 0, %pi/2);

(%o2) 1

(%i3) integrate(exp(-x^2), x, 0, inf);

(%o3)

√
%pi

2� �

第8章 関数のグラフ

8.1 １変数関数のグラフ
１変数の関数のグラフは wxplot2dまたは plot2dを用いて描くことができ

ます．wxplot2dを用いると，グラフがインラインで描かれます．一方，plot2d

を用いると，デフォルトでは別プログラムの gnuplotが立ち上がり，別の窓

にグラフを描いてくれます．gnuplot以外にも openmathを用いてグラフを

表示することもできます．その場合は [plot_format, openmath]というオ

プションを付けることになります．

たとえば y = sinxのグラフを −2π ≤ x ≤ 2π の範囲で描くには，次の３

種類の方法があります．

wxplot2d(sin(x), [x,-2*%pi,2*%pi]);

plot2d(sin(x), [x,-2*%pi,2*%pi]);

plot2d(sin(x), [x,-2*%pi,2*%pi], [plot_format, openmath]);

openmathで表示するとき，いちいち{[plot_format, openmath] という

オプションを付けるのが面倒だと思うなら，

set_plot_option([plot_format, openmath]);

としておくと，それ以後はオプションを指定しなくても openmathで表示され

ます．gnuplotに戻す場合は，set_plot_option([plot_format, gnuplot]

とします．

ちなみに，3Dグラフィックスを表示する場合は openmathが圧倒的にきれ

いですが，2Dグラフィックスではさほどの違いはありません．

同じことなので，以下では，gnuplotによる方法を用いて記述します．

y = sinxのグラフを表示したとき，見慣れたサインカーブとは異なってい

ると思われたかもしれません．それは x軸と y軸のスケールが異なっている

ためです．wxplot2dや gnuplotによる plot2dでは [gnuplot preamble,

"set size ratio -1"] というオプションを付けてやると x 軸と y 軸のス

ケールを同じに表示してくれます．� �
(%i1) plot2d(sin(x), [x,-2*%pi,2*%pi], [gnuplot preamble,

"set size ratio -1"]);� �

入力窓の下にある Plot 2D.. というボタンをクリックして関数を描くことも

できます．ボタンをクリックするとPlot 2D窓が開きますので，Expression(s):

に sin(x)と入力し，2つある Variable:の上のほうが xになっていることを

確認し，From:に-2*%pi を To:に 2*%pi を入力し OK をクリックすると同

じグラフが表示されます．この窓を用いると，Format:で inline, gnuplot,

openmathを選ぶことができ，Options:に [gnuplot preamble, "set size

ratio -1"] のようにオプションを指定することができます．このようにし

ておくと，一度指定したオプションを記憶しておいてくれるので，何度も利

用したいときには便利です．

Ticks:の値を変えることにより，nticksを変更できます．グラフによって

は，ぎざぎざが大きくきれいに表示されないときがありますが，それは描画

するために用いられる点の数が少なすぎるためです．そこで，plot2dのオプ

ションで nticksの値を大きく指定し，描画の点を増やすときれいになりま

す．ちなみに，nticksのディフルト値は 10です．

Plot 2D窓でなく，インラインで，nticksの値を 20にするには，たとえば

plot2d(sin(x), [x,-2*\%pi,2*\%pi], [nticks, 20]);

のようにします．

複数の関数を一緒に表示してみます．たとえば y = sinxと y = cosxのグ

ラフを −2π ≤ x ≤ 2πの範囲で描くには，� �
(%i2) plot2d([sin(x),cos(x)], [x,-2*%pi,2*%pi]);� �

のように入力します．先ほどの Plot 2D窓を用いる場合は，Expression(s)に

sin(x),cos(x)と 2つ並べて入力します．

8.2 ２変数関数のグラフ
2変数の関数を描く場合は wxplot3dまたは plot3dを使います．1変数のと

きと同様に wxplot3dを用いると，グラフがインラインで描かれ，plot3dを

用いると別の窓に描かれます．また，別窓で描く場合，gnuplotや openmath

を用いることができることも同様です．3Dグラフィックスでは，plot3dを

用いると，表示されたグラフを回転させることができます．しかし wxplot3d

を用いるとインラインで表示されるためグラフを回転させることができませ

ん．そのため，plot3dの方がベターといえます．

たとえば，z = sin(x2 + y2)を 0 ≤ x ≤ π/2, 0 ≤ y ≤ π/2の範囲で描くに

は次のようにします．

� �
(%i3) plot3d(sin(x^2+y^2), [x,0,%pi/2], [y,0,%pi/2]);� �
また，1変数のときのように，入力窓の下にある Plot 3D.. をクリックして

も同様のことができます．ただし，描画する点の個数を指定するには，先ほ

どの Ticks:ではなく Grid:の値を変更します．

8.3 応用
Maximaを用いて高校数学の問題を解いてみましょう．

例題. y = x2 と y = 2x+ 5で囲まれる部分の面積を求めよ．

まず f(x) = x2, g(x) = 2x+ 5とおき，この２つの関数のグラフの交点の

座標を求めます．� �
(%i4) f(x):=x^2$ g(x):=2*x+5$

(%i6) solve(g(x)-f(x)=0,x);

(%o6) [x = 1−
√
6, x =

√
6 + 1 ]� �

交点の x座標は 1−
√
6と

√
6+1であることが分かりましたが，このままで

は扱いにくいので，上のリストから x=という部分を削除して，[1−
√
6,
√
6+1]

というリストを作り kaiという変数に代入することにします．

一般に式 a=bから右辺 bを取り出すのは rhs(a=b)とします．また，左辺

は lhs(a=b)です．しかし，この場合は rhs(%)のようにしてもうまくいきま

せん．それは出力 (%o6) がリストであるためで，このような場合は mapとい

う命令を用いて次のようにします．� �
(%i7) kai:map(rhs,%);

(%o7) [ 1−
√
6,
√
6 + 1 ]� �

そこで，f(x)と g(x)のグラフを同じ平面上に描いてみます．xの範囲は

交点の，x = 1−
√
6と x =

√
6 + 1が含まれるように選びます．とりあえず

−5 ≤ x ≤ 5の範囲にして見ます．� �
(%i8) plot2d([f(x),g(x)], [x,-5,5]);� �

図を見ると，２つの曲線で囲まれる部分では g(x) ≥ f(x) なので，面積は

g(x)− f(x)を 1−
√
6から

√
6 + 1まで積分することによって求まることが

わかります．� �
(%i9) integrate(g(x)-f(x),x, kai[1], kai[2]);

(%o9)
12
√
6 + 17

3
+

12
√
6− 17

3

(%i10) factor(%)

(%o10) 8
√
6� �

以上で，面積が 8
√
6であることがわかりました．

演習問題
1. 関数 f(x) = 1− |2x− 1| (0 ≤ x ≤ 1)に対して，f , f2, f3, f4のグラフ

を，１つの座標平面に描け．ただし fnは f の n回の合成関数 f ◦ · · · ◦ f︸ ︷︷ ︸
n

を表す．また，次で与えられる関数 gのグラフを描け．

g =
f

2
+

f2

22
+ · · ·+ f10

210

27



注意：(1) f , f2, f3などは，たとえば f[1], f[2], f[3]などのように，

リストを用いて定義してやるとよいかもしれない．具体的には，次のよ

うに帰納的に定義する．

f[1](x):=1-abs(2*x-1); f[n](x):=f[1](f[n-1](x));

(2) 出来上がりがきれいになるように，plot2dのオプションをうまく

設定して書くこと．

(3) 次の無眼級数により定義される関数 T は高木関数とよばれている．

T =
f

2
+

f2

22
+ · · ·+ fn

2n
+ · · ·

この関数は，区間 [0, 1]で連続であるが，いたるところ微分不可能になっ

ている．また，この関数のグラフはフラクタルの一つの例になっている．

2. 2つの曲線 y = x3 − 4x, y = −x2 + 4で囲まれる領域の面積を求めよ．

3. 関数 f(x) = x sin x
2 を考える．点 (π, f(π))における曲線 y = f(x)の接

線 ℓの方程式を求めよ．また ℓとこの曲線 y = f(x)とで囲まれた部分

の面積を求めよ．（愛媛大学 1999年度入試問題）

4. kを正の定数とし，g(x)を連続な関数とする．関数 h(x)を

h(x) =

∫ x

0

g(t)ek(x−t) dt

とすると，h′(x) = kh(x) + g(x)であることを示せ．また，f(x)を

f(x) = sin kx+
1

k

∫ x

0

g(t) sin(kx− kt) dt

としたとき，f ′′(x)を f(x)及び g(x)を用いて表せ．（山口大学 1999年

度入試問題）

第9章 行列と行列式

9.1 行列と行列式の計算
行列は matrixを用いて定義します．行列

P =

0 1 1

1 0 1

1 1 0

 , Q =

 1 −1 2

−2 3 −1

−1 4 7

 , R =

[
1 2 3

4 5 6

]

を定義して，R(2P −Q)を計算してみます．行列のスカラー倍は数の積と同

じ ∗を用いますが，行列の積にはピリオッドを用います．� �
(%i1) P:matrix([0,1,1],[1,0,1],[1,1,0]);

(%o1)

0 1 1

1 0 1

1 1 0


(%i2) Q:matrix([1,-1,2],[-2,3,-1],[-1,4,7]);

(%o2)

 1 −1 2

−2 3 −1

−1 4 7


(%i3) R: matrix([1,2,3],[4,5,6]);

(%o3)

[
1 2 3

4 5 6

]
(%i4) R.(2*P-Q);

(%o4)

[
16 −9 −15

34 −15 −27

]
� �
行列 Aの n乗は，A^nでなく A^^nとして求めます．ちなみに A^nとする

と，各成分が n乗されてしまいます．

� �
(%i5) A:matrix([1,2,3],[-1,2,-5],[0,2,0]);

(%o5)

 1 2 3

−1 2 −5

0 2 0


(%i6) A^^4;

(%o6)

−21 −136 −79

53 −24 255

10 −96 62


(%i7) A^4;

(%o7)

1 16 81

1 16 625

0 16 0


� �
行列の列ベクトルや行ベクトルは colや rowを用いて取り出すことができ

ます．また addcolや addrowで行列に列や行を追加することができます．� �
(%i8) B:matrix([2,-3,-1],[3,3,10]);

(%o8)

[
2 −3 −1

3 3 10

]
(%i9) u:col(B,2);

(%o9)

[
−3

3

]
(%i10) row(B,1);

(%o10)
[
2 −3 −1

]
(%i11) C:addcol(B,u,matrix([0],[1]));

(%o11)

[
2 −3 −1 −3 0

3 3 10 3 1

]
(%i12) addrow(C,[1,2,3,4,5]);

(%o12)

2 −3 −1 −3 0

3 3 10 3 1

1 2 3 4 5


� �
identは単位行列を，zeromatrixは零行列を生成します．

� �
(%i13) ident(4);

(%o13)


1 0 0 0

0 1 0 0

0 0 1 0

0 0 0 1


(%i14) zeromatrix(2,3);

(%o14)

[
0 0 0

0 0 0

]
� �
ematrixはある成分ひとつだけ値を持ち，他はすべて 0である行列を生成

します．たとえば，(2, 3)成分が −4である 3× 5行列を生成するには次のよ

うにします．� �
(%i15) ematrix(3,5,-4,2,3);

(%o15)

0 0 0 0 0

0 0 −4 0 0

0 0 0 0 0


� �
対角成分がすべて同じ対角行列は diagmatrixを用いて簡単に与えること

ができます．� �
(%i16) diagmatrix(3,5);

(%o16)

5 0 0

0 5 0

0 0 5


� �
行列の各成分が関数で与えられる様な場合は genmatrixを用いて与えま

す． たとえば (i, j)成分が i+ jであるような 2× 3行列は次のようにして与

えます．� �
(%i17) h[i,j]:=i+j;

(%o17) hi,j := i+ j

(%i18) genmatrix(h,2,3);

(%o18)

[
2 3 4

3 4 5

]
� �

(%i17)で hを定義するとき，関数の定義の h(i,j):=i+j ではなく，リス

トの定義の h[i,j]:=i+jであることに注意してください．

今度は，先ほど定義した行列 P , Qの逆行列を求めてみます．逆行列を求

めるには invertを用います．� �
(%i19) invert(P); invert(Q);

(%o19)


−1

2

1

2

1

2
1

2
−1

2

1

2
1

2

1

2
−1

2


Division by 0

- - an error. Quitting. To debug this try debugmode(true);� �
Qの逆行列を求めるのは失敗しました．そこで P, Qの行列式を調べてみ

ます．行列式には determinantを用います．� �
(%i21) determinant(P); determinant(Q);

(%o21) 2

(%o22) 0� �
P の行列式は 2であるため P は正則行列であることが分かります．一方Q

の行列式は 0であり，Qは逆行列を持たないことが分かります．

転置行列は transposeを用います．ここでは，P , Q, Rの転置行列を，map

を用いていっぺんに求めてみます．� �
(%i23) map(transpose,[P,Q,R]);

(%o23) [

0 1 1

1 0 1

1 1 0

 ,

 1 −2 −1

−1 3 4

2 −1 7

 ,

1 4

2 5

3 6

 ]

� �
この mapは与えられた関数の複数の変数に対しての値を求めるとき便利で

す．たとえば，次のようにして用います．� �
(%i24) f(x):=x^2$

(%i25) map(f,[1,2,3]);

(%o25) [1, 4, 9]� �

数ベクトルは，列がただ一つの特別な行列と考えることができるので，た

とえばベクトル

 1

0

−1

は matrix([1],[0],[-1])として与えることができ

ます．しかし，もっと簡単にリスト [1,0,-1]の転置行列として与えること

もできます．� �
(%i26) v:transpose([1,0,-1]);

(%o26)

 1

0

−1


� �
連立一次方程式の係数行列は coefmatrixで得られます．また，拡大係数

行列は augcoefmatrixで得られます．� �
(%i27) eq:[2*x-3*y-1=0,3*x+3*y=10];

(%o27) [−3y + 2x− 1 = 0, 3y + 3x = 10]

(%i28) coefmatrix(eq,[x,y]);

(%o28)

[
2 −3

3 3

]
(%i29) augcoefmatrix(eq,[x,y]);

(%o29)

[
2 −3 −1

3 3 −10

]
� �
実を言うと，正確には augcoefmatrixは拡大係数行列にはなっていませ

ん．上の例だと，拡大係数行列は

[
2 −3 1

3 3 10

]
でなくてはなりません．で

もいいでしょう．

正方行列の余因子行列を求めるには adjointを用います．

� �
(%i30) B:matrix([1,2,3],[0,5,6],[7,8,9]);

(%o30)

1 2 3

0 5 6

7 8 9


(%i31) adjoint(B);

(%o31)

 −3 6 −3

42 −12 −6

−35 6 5


� �
行列の階数を求めるには rankを用います．� �
(%i32) rank(B);

(%o32) 3� �
トレースは mattraceを用いますが，この場合は前もって nchrplをロー

ドしておく必要があります．� �
(%i33) load("nchrpl");

(%o33) C:/PROGRA˜1/MAXIMA˜1.1/share/maxima/5.17.1/share/matrix/nchrpl.mac

(%i34) mattrace(B);

(%o34) 15� �
今度は行列 P の固有値および固有ベクトルを求めましょう．ここで P の固

有値 λと，λの固有ベクトル v ( ̸= 0)とは

Pv = λv

を満たすものです．

例えば，先ほどの行列 P =

0 1 1

1 0 1

1 1 0

 とベクトル v =

 1

0

−1

 に対して，
次のように Pv = (−1)v となるのがわかるので，−1は P の固有値で，v は

固有値 −1に対する固有ベクトルとなっていることがわかります．

34



� �
(%i35) P.v;

(%o35)

−1

0

1


� �
固有値を求めるためには，最初に eigenというマクロを読み込む必要があ

ります．� �
(%i36) load("eigen");

(%o36) C:/PROGRA˜1/MAXIMA˜1.1/share/maxima/5.17.1/share/matrix/eigen.mac

(%i37) eivects(P);

(%o37) [ [ [ 2,−1 ], [ 1, 2 ] ], [ 1, 1, 1 ], [ 1, 0,−1 ], [ 0, 1,−1 ] ]� �
最後の出力について説明します．出力は，第１要素が [ [ 2,-1 ],[ 1,2 ] ], 第２要

素から第４要素が [ 1,1,1 ], [ 1,0,-1 ], [ 0,1,-1 ] のリストになっています．各要

素はリストになっていますが，第１要素は他の要素と形が違って，リストを

要素としたリストになっています．一般に出力の第１要素は

[[e1, e2, . . . , ek], [m1,m2, . . . ,mk]]

という形をしており，eiが固有値でmiは固有値 eiの重複度を表しています．

第２要素以降が固有空間の基底である固有ベクトルの列になり，最初の m1

個は e1 の固有空間の基底，次のm2 個が e2 の固有空間の基底といった具合

です．

上の例では，最初の要素から，2と−1が固有値であり，2の重複度は 1, −1

の重複度は 2であることが分かります．

その次から，ベクトル

11
1

 が固有値 2に対する１つの固有ベクトルであ

り，ベクトル

 1

0

−1

,
 0

1

−1

 が固有値−1に対する線形独立な２つの固有ベク

トルになることが分かります．

そこでP を対角化しましょう．一般に n次正方行列Aの固有値が e1, . . . , en

で，それらに関する固有ベクトルの線形独立な組 v1, . . . , vn が与えられたと

き，これらのベクトルを 列とする行列X = [v1 . . . vn]により

X−1AX =


e1 0 . . . 0

0 e2 . . . 0
...

...
. . .

...

0 0 · · · en


となることがわかります．

今の場合は

X =

1 1 0

1 0 1

1 −1 −1

 とすると X−1PX =

2 0 0

0 −1 0

0 0 −1


が得られることになります．

まず，eivects(P)の出力から，固有ベクトルだけからなるリストを作り，

それを xとします．リストの１番目を削除したリストは restを用いて求め

ることができます．� �
(%i38) x:rest(%);

(%o38) [ [ 1, 1, 1 ], [ 1, 0,−1 ], [ 0, 1,−1 ] ]� �
次に xの各要素を各行にした行列を考え，その転置行列をX とし，X−1PX

を計算します．� �
(%i39) X:transpose(matrix(x[1],x[2],x[3]));

(%o39)

1 1 0

1 0 1

1 −1 −1


(%i40) invert(X).P.X;

(%o40)

2 0 0

0 −1 0

0 0 −1


� �
固有値を対角成分とする対角行列が得られました．

これを用いて Pn を次のようにして計算することができます．

Pn = (XX−1)P (XX−1)P · · ·P (XX−1)

= X(X−1PX)nX−1

= X

2
n 0 0

0 (−1)n 0

0 0 (−1)n

X−1

Maximaでは行列 Aに対して A^n は行列の n乗でなく，各成分の n乗を与

えます．ここで出力 (%o40)は対角行列なので，その n乗は各成分の n乗に

なるので，次のようにして Pn が求まります．� �
(%i41) %^n;

(%o41)

2
n 0 0

0 (−1)n 0

0 0 (−1)n


(%i42) X.%.invert(X);

(%o42)


2n

3
+

2(−1)n

3

2n

3
− (−1)n

3

2n

3
− (−1)n

3
2n

3
− (−1)n

3

2n

3
+

2(−1)n

3

2n

3
− (−1)n

3
2n

3
− (−1)n

3

2n

3
− (−1)n

3

2n

3
+

2(−1)n

3


(%i43) factor(%)

(%o43)


2n + 2(−1)n

3

2n − (−1)n

3

2n − (−1)n

3
2n − (−1)n

3

2n + 2(−1)n

3

2n − (−1)n

3
2n − (−1)n

3

2n − (−1)n

3

2n + 2(−1)n

3


� �
9.2 応用
漸化式で与えられる数列の一般項を行列を用いて解いてみます．フィボナッ

チ数列と呼ばれる有名な数列があります．これは

f1 = f2 = 1, fn+2 = fn + fn+1 (n ≥ 1)

で与えられる数列 {fn} です．ここで２次元ベクトル vn を vn =

[
fn+1

fn

]

(n ≥ 1)で定めます．このとき，行列 A =

[
1 1

1 0

]
を考えると，条件式は次

のようになります．

v1 =

[
f2

f1

]
=

[
1

1

]
, Avn−1 =

[
1 1

1 0

][
fn

fn−1

]
=

[
fn + fn−1

fn

]
=

[
fn+1

fn

]
= vn

よって，上のようにして An−1 を求めると，一般項 fn は

vn = An−1v1

から求まります．

演習問題

1. 行列A =

a 1 0

0 a 1

0 0 a

に対して，An =

an nan−1 n(n− 1)an−2

2
0 an nan−1

0 0 an


であることを帰納法で示せ．

ヒント：

an nan−1 n(n− 1)an−2

2
0 an nan−1

0 0 an

を nの関数として定義します．

つまり行列 A(n)を次のように定義します．

A(n):=matrix([a^n,n*a^(n-1),n*(n-1)*a^(n-2)/2],[0,a^n,n*a^(n-1)],[0,0,a^n])

このとき An = A(n)を帰納的に示せばよいのです．

2. f : R2 → R2 を

f

(
x

y

)
=

(
x cos t− y sin t

x sin t+ y cos t

)

と定める．このとき，R2 の標準基底

(
1

0

)
,

(
0

1

)
に関する f の表現行

列M を求めよ．またMn =

(
cosnt − sinnt

sinnt cosnt

)
(n ≥ 1) となることを

帰納法で示せ．

ヒント：三角関数を含む式を加法定理を用いてただ一つの sinまたは cos

のみを含む式の和として表すには trigreduceを用います．たとえば

trigreduce(2*sin(t)*cos(t))とすると，sin(2t)が帰ってきます．

第10章 平面図形

10.1 陽関数のグラフ
平面図形を描くには，関数のグラフと同様に wxplot2dまたは plot2dを

用います．ここでも gnuplotを用いて記述します．

関数 y = f(x), y = g(x), . . . のグラフを a ≤ x ≤ bの範囲で描くには，す

でにやったように

plot2d([f(x),g(x),...],[x,a,b])

のようにします．しかし，必ずしもうまくいかないことがあります．たとえ

ば，y =
1

x
を −5 ≤ x ≤ 5の範囲で描いてみます．

� �
(%i1) plot2d(1/x,[x,-5,5]);� �
表示されるグラフはめちゃめちゃ変になったと思います．これは xが 0に

近づくと，関数 y = 1/x の値の絶対値が限りなく大きくなる（つまり直線

x = 0がこのグラフの漸近線になる）ためです．実際，グラフをよく見ると，

画面に現れている x軸の範囲と y軸の範囲が大きく異なっていることが分か

ります．そこで，今度は yの範囲も指定してやると，そこからはみ出す部分

は描画しなくなり，きれいなグラフが表示できます．� �
(%i2) plot2d(1/x,[x,-5,5],[y,-5,5]);� �
ちなみに，Plot 2D窓を用いる場合は，下の Variable: で y の範囲を指定

します．

10.2 媒介変数表示される曲線
媒介変数表示 (x(t), y(t)) (a ≤ t ≤ b)で与えられている平面曲線を描くには

plot2d([parametric,x(t),y(t)],[t,a,b]);

のように parametricというオプションを用います．

半径1の円を描いてみます．半径1の円は，媒介変数表示で (x, y) = (cos t, sin t)

(0 ≤ t ≤ 2π)なので，次のようにしてみます．

� �
(%i3) plot2d([parametric,cos(t),sin(t)],[t,0,2*%pi]);� �
Plot 2D窓を用いる場合は，Expression(s): に，そのままずらずら

parametric,cos(t),sin(t)

と書き込めばよいです．

上の円は，よく見るときちんとした円でなく，多少ぎざぎざの曲線で結ば

れていることがわかります．これは以前にも書きましたが，Maximaが区間

0 ≤ t ≤ 2π をいくつかの点で等分し，各分点の値を代入してできる平面上

の点を順番に結ぶことにより図形を描いているからです．区間の分割数は，

nticksという変数に格納されていますので，nticksの値を大きくしてやる

と（たとえば 300くらいに），よりきれいな円が描かれることになります．

逆に nticksの値を小さくすると，正多角形を描くことができます．つま

り正 n角形を描くには nticksを n + 1にして円を表示すればよいことにな

ります．たとえば，正 5角形なら� �
(%i4) plot2d([parametric,cos(t),sin(t)],[t,0,2*%pi],[nticks,

6]);� �
のようにします．

さらに，tの範囲を 0 ≤ t ≤ 4πにすると，今度は星型が表示されます．� �
(%i5) plot2d([parametric,cos(t),sin(t)],[t,0,4*%pi],[nticks,

6]);� �
一般に nとmが互いに素な自然数のとき，nticksを n+1にして 0 ≤ t ≤

2mπの範囲で描いてできる星型を正 n/m角形といいます．上の星型は正 5/2

角形です．

通常の陽関数のグラフとパラメータオプションの図の混じった描画も可能

です．たとえば，y = x3 +2 (−3 ≤ x ≤ 3)と単位円を同時に描きたいときは

次のようにします．� �
(%i6) plot2d ([x^3 + 2, [parametric, cos(t), sin(t)]],

[x, -3, 3], [y, -3, 3], [t, 0, 2*%pi];� �

上の例では nticksを共通にとっていますが，複数のパラメーターオプショ

ンの図で，パラメーターの範囲や nticksの値を個別に設定したいときは

plot2d ([[parametric, x(t), y(t), [nticks, n], [t, a, b]],

[parametric, x’(t), y’(t), [nticks, n’],[t, a’, b’]]]);

のようにします．

最後にスピノグラフを描いて見ましょう．� �
(%i7) a:96$ b:45$

(%i9) load("functs")$

(%i10) c:lcm(a,b);

(%o10) 1440

(%i11) A:a/(2*%pi)$ B:b/(2*%pi)$ C:B*0.8$

(%i14) plot2d([parametric, (A-B)*cos(t)-C*cos((a-b)/b*t),

(A-B)*sin(t)+C*sin((a-b)/b*t)], [t,0,2*%pi*c/a],[nticks,

1000]);� �
演習問題

1. サイクロイドを描け．サイクロイドとは，円が直線の上を滑らずに転が

るときに，円の周上の１点が描く軌跡をいう．転がる円の半径が 1のと

き，サイクロイドは tを媒介変数として (x, y) = (t− sin t, 1− cos t)で

与えられる．

2. アステロイド：(x, y) = (cos3 t, sin3 t) (0 ≤ t ≤ 2π)を描け．

3. 極方程式 r = cos 3t (0 ≤ t ≤ 2π)で与えられる三葉線を描け．また，

r = cos 2t (0 ≤ t ≤ 2π)で与えられる四葉線を描け．ただし極方程式

r = r(t) (a ≤ t ≤ b)で与えられる曲線とは，{ (r(t) cos t, r(t) sin t) |
a ≤ t ≤ b }と媒介変数表示される曲線である．

4. 極方程式 r = 1 + cos t (0 ≤ t ≤ 2π)で与えられるカージオイドを描け．

5. 極方程式 r = esin t − 2 cos(4t) + sin(t/12) (0 ≤ t ≤ 24π)で与えられる

フェイの蝶を描け．

6. 頂点を共有する正 5角形と正 5/2角形を同一の平面に描け．

7. 平面内のx軸上の線分{ (x, 0) | 1 ≤ x ≤ 3 }を，点 (2, 0)を中心に角2π/7

だけ回転した線分を，正方形の領域 {(x, y) | −4 ≤ x ≤ 4, −4 ≤ y ≤ 4}
内に描け．

第11章 空間図形

11.1 陽関数のグラフ
空間図形を描くには wxplot3dまたは plot3dを用いますが，表示した図

形を回転させることができる plot3dがベターです．関数 z = f(x, y)のグラ

フを a ≤ x ≤ b, c ≤ y ≤ dの範囲で描くには，すでにやったように

plot3d(f(x,y),[x,a,b],[y,c,d])

のようにします．ただし，平面図形のように複数の図形を同時に表示するこ

とはできません．めちゃめちゃ残念です．

関数 z =
x2y

x4 + y2
を −1 ≤ x ≤ 1, −1 ≤ y ≤ 1 の範囲で描くには次のよう

にします．� �
(%i1) plot3d(x^2*y/(x^4+y^2),[x,-1,1],[y,-1,1]);� �
図形のぎざぎざが気になるときは，すでに書いたように gridというオプ

ションを大きくしてやります．ちなみに gridの初期値は 30× 30です．� �
(%i2) plot3d(x^2*y/(x^4+y^2),[x,-1,1],[y,-1,1],[grid,50,50]);

� �
11.2 媒介変数表示される曲面
媒介変数 (x(s, t), y(s, t), z(s, t)) (a ≤ s ≤ b, c ≤ t ≤ d)で与えられている

曲面を描くには，単に

plot3d([x(s,t),y(s,t),z(s,t)],[s,a,b],[t,c,d]);

のようにします．平面曲線のときのように parametricというオプションは

必要ありません．たとえば，半径 1の球面は，媒介変数で

(x, y, z) = (cos s cos t, cos s sin t, sin s) (−π/2 ≤ s ≤ π/2, 0 ≤ t ≤ 2π)

なので次のようになります．

� �
(%i3) plot3d([cos(s)*cos(t),cos(s)*sin(t),sin(s)],[s,-%pi/2,%pi/2],

[t,0,2*%pi]);� �
平面図形のときと同様に gridの値を小さくすると，多面体が描けます．残念

ながら，正多面体になるわけではありませんが，たとえば球面を [grid,2,4]

で描くと正八面体になります．� �
(%i4) plot3d([cos(s)*cos(t),cos(s)*sin(t),sin(s)],[s,-%pi/2,%pi/2],

[t,0,2*%pi],[grid, 2, 4]);� �
演習問題

1. xz平面の線分 { (3, 0, t) | −1 ≤ t ≤ 1 }を z軸を中心に１回転したとき

できる曲面を描け．この曲面をアニュラスとよぶ．

2. 次をヒントにしてトーラスを描け．

(a) まず，x 軸上の点 (3, 0, 0) を中心に xz 平面内で半径 1 の円を考

える．

(b) この円を媒介変数表示で (x(t), y(t), z(t)) (a ≤ t ≤ b) のように

表す．

(c) この円を 1のように z 軸を中心に１回転したとき，その軌跡とし

て得られるのがトーラスである．

3. 次をヒントにしてメビウスの帯を描け．トーラスを描いたときと同様に

考えると，xz 平面内の線分を線分の中心を軸に自転させながら，z 軸

を中心に１回転させてやる．その際 z 軸中心に１回転する間に，線分

が 1/2回転自転するようにすれば，出来上がりはメビウスの帯になる．

より詳しくは，次のようにする．

(a) まず，xz 平面の長さ 1の線分で (3, 0, 0)を中点とし，x軸とのな

す角が θ (0 ≤ θ ≤ π) であるものを lθ とする．

(b) lθを媒介変数表示で (xθ(t), yθ(t), zθ(t)) (a ≤ t ≤ b)のように表す．

(c) lθ を z 軸を中心に角 2θ回転して得られる線分を l′θ とすると，線

分の束 { l′θ | 0 ≤ θ ≤ π }全体からなる曲面がメビウスの帯になる．

第12章 プログラム

12.1 条件分岐
Maximaでは条件分岐のための if文が用意されています．書式は

if 条件式 then 真の場合の処理 else 偽の場合の処理;

のようになります．条件式には，等号や不等号 =, <,>,>=, <=，あるいは

等しくないかを問う#などを使うことができます．

次のような関数 f(x)を考えます．

f(x) =

x2 x > 0のとき

0 x ≤ 0のとき

この関数は，次のように条件分岐を使って定義できます．� �
(%i1) f(x):=if x>0 then x^2 else 0;� �
条件分岐は入れ子にもできます．たとえば，上の関数を次のように変更し

たいとします．

f(x) =


x2 x > 0のとき

0 0 ≥ x > −2のとき

x3 + 8 x ≤ −2のとき

この場合，次のように定義します．� �
(%i2) f(x):=if x>0 then x^2 else (if x> -2 then 0 else

x^3+8);� �
関数を再帰的に定義することもできます．たとえば，f(x) = x!を定義する

には次のようにします．

� �
(%i3) f(x):=if x=0 then 1 else x*f(x-1);

(%o3) f(x) := if x = 0 then 1 else xf(x− 1)

(%i4) f(5);

(%o4) 120� �
上では，x = 0に対し f(0) = 0! = 1と定義し，x > 1に対しては x! =

x× (x− 1)!を用いて，f(x) = xf(x− 1)で定義しています．

ある数が正または 0であるかどうかをチェックするためのプログラム sei

を作ってみます．� �
(%i5) sei(x):= if x > 0 then "正の数" else (if x=0 then "

ゼロ" else "負の数");

(%i6) sei(0.5);

(%o6) 正の数

(%i7) sei(0);

(%o7) ゼロ

(%i8) sei(-3);

(%o8) 負の数� �
上のプログラムで引用符”で囲まれた”正の数”などは，引用符内の文字（こ

の場合は正の数）を印字させるための表現です．

等号や不等号以外にも，変数の属性を問うための関数を用いることができ

ます．変数が整数かを問う integerp，偶数かをを問う evenp，奇数かを問

う oddp，素数かを問う primep，有理数かを問う ratnump，リストかを問う

listp，行列かを問う matrixpなどを用いることもできます．

たとえば，引数として与えられた正の整数が素数かどうかを判定するため

に用いられる関数 primepを用いてみます．� �
(%i9) primep(2);

(%o9) true

(%i10) primep(4);

(%o10) false� �
自然数 nが与えられたとき，n以下の素数の個数を求める関数 primenを

作ってみます．

� �
(%i11) primen(n):=if n=2 then 1 else (if primep(n) then

primen(n-1)+1 else primen(n-1))$

(%i12) primen(100)

(%o12) 25

(%i13) primen(1000)

Maxima encountered a Lisp error:

Error in PROGN [or a callee]: Bind stack overflow.

Automatically continuing.

To reenable the Lisp debugger set *debugger-hook* to nil.� �
100以下の素数の個数は 25個であることが分かりますが，次に 1000以下の

素数の数を求めようとしてエラーになりました．スタックがオーバーフロー

した様です．

xと x+ 2が共に素数であるとき，これらを双子素数といいます．そこで，

数 xが与えられたとき xと x+ 2が双子素数かどうかを問う関数 twinprime

は次の様に定めることができます．� �
(%i14) twinprime(x):=if primep(x) and primep(x+2) then

print(x,"と",x+2,"は双子素数") else print

(x,"または",x+2,"は素数ではない");

(%o14) twinprime(x):=if primep(x) and primep(x+ 2) then print

(x,と,x+ 2,は双子素数) else print(x,または,x+ 2,は素数ではない)

(%i15) twinprime(3)$

3 と 5 は双子素数

(%i16) twinprime(7)$

7 または 9 は素数ではない

(%i17) twinprime(11777)$

11777 と 11779 は双子素数� �
上の例で twinprime(7)$のように$を用いていますが，$を用いないと次のよ

うになります．� �
(%i18) twinprime(3);

3 と 5 は双子素数

(%o18) は双子素数� �

12.2 繰り返し操作
繰り返し操作のためには for文が用意されています．書式は

for カウンタ名:初期値 step 増分 thru 終了値 do (反復実行手続き);

for カウンタ名:初期値 step 増分 while 条件式 do (反復実行手続き);

の２種類があります．この stepは省略することができ，その場合は 1が指

定されたものとなります．そして，doの括弧の中身は，複数の作業を指定す

るときは，順番にコンマで区切って列挙します．

1000以下の素数の個数を繰り返し操作を用いて求めてみましょう．� �
(%i19) j:0;

(%o19) 0

(%i20) for i:2 thru 1000 do (if primep(i) then j:j+1);

(%o20) done

(%i21) j;

(%o21) 168� �
今度は 1000以下の素数の個数も，きちんと 168個と計算できたようです．

方程式を数値計算によって解くための反復法によるアルゴリズムの 1つに

ニュートン法と呼ばれるものがあります．これは，f(x) = 0で与えられる方

程式に対して，解の近くの値 x0 を適当に決めてやり，漸化式

xn+1 = xn − f(xn)

f ′(xn)
(n ≥ 1)

により数列 xnを定めると，この数列が f(x) = 0の解に収束することが多い

ということによります．よって，十分大きな nをとれば，xn は f(x) = 0の

解の一つの近似値を与えることになります．

たとえば，f(x) = x2 − 2 = 0の解の近似値をニュートン法で求めてみま

す．（もちろんこの解は
√
2です．）

� �
(%i22) f(x):=x^2-2$

(%i23) define(g(x),x-f(x)/diff(f(x),x));

(%o23) g(x) :=
x2 − 2

2x
(%i24) a:4;

(%o24) 4

(%i25) for i:1 thru 10 do(a:g(a));

(%o25) done

(%i26) a,numer;

(%o26) 1.414213562373095

(%i27) a:-1;

(%o27) −1

(%i28) for i:1 thru 10 do(a:g(a));

(%o28) done

(%i29) a,numer;

(%o29) −1.414213562373095� �
関数の定義に defineを用いていることに注意してください．

上の例で，初期値を4にとると，
√
2の近似値である1.414213562373095を得

ることができ，−1にすると，今度は−
√
2の近似値である−1.414213562373095

を得ます．

12.3 ブロック化
既存の関数に条件分岐や繰り返し操作を加えると，様々な処理を行うこと

ができます．れらの手順をまとめて一つの関数にする方法がこのブロック化

です．ブロック化するときの書式は，

block([局所変数のリスト], 一連の手続き, return(計算結果));

のようになります．

前節の繰り返し操作を用いた素数の個数の求め方を一般化して，n以下の

素数の個数を求める関数 primen2を作ってみましょう．� �
(%i30) primen2(n):=block([j:0], for i:2 thru n do (if

primep(i) then j:j+1), return(j))$

(%i31) primen2(100)

(%o31) 25

(%i32) primen2(1000)

(%o32) 168� �

上のプログラムでは blockの中で，まず局所変数 jを与え，その初期値と

して 0を代入しています．その後に，2から nまでの整数 iが素数かどうか

をチェックし，素数であれば jに 1を加えてゆき，最後まで調べ終わったら，

jの値を返しています．

あるクラスの全員の試験の点数のリストが与えられたとき，合格者の人数

を求める関数 goukakuを作ってみます．ここで，成績のリストとは，全員の

点数を並べた [70,45,82,100,13,0,68,93,48] のようなものとします� �
(%i33) goukaku(lis):= block([ninzuu:0], for i:1 thru

length(lis) do (if lis[i] >= 60 then ninzuu:ninzuu+1),

return(ninzuu))$

(%i34) goukaku([70,45,82,100,13,0,68,93,48]);

(%o34) 5� �
上のプログラムでは blockの中で，まず局所変数 ninzuuを与え，初期値

として 0を代入しています．その後に，リスト lisの要素を１番目から順に

60以上かを調べ，60以上の点数があるたびに，変数 ninzuuに 1を加えてゆ

き，最後まで調べ終わったら，最後に ninzuuの値を返しています．ちなみ

に lengthはリストの要素の数を与える関数です．

演習問題
1. 自然数 nに対する値 f(n)がフィボナッチ数列の第 n項を与える関数

f を定義し，初項から第 20項までのリストを作れ．ここでフィボナッ

チ数の第 n 項 f(n) は，n = 1, 2 のとき，f(n) = 1，n ≥ 3 のとき，

f(n) = f(n− 1) + f(n− 2)で与えられる．

2. 自然数 nが与えられたとき，n番目の素数を求める関数 nprimeを作れ．

また，1000番目の素数を求めよ．

3. すべての正の整数上で定義された関数 f(n)を次で定める．

f(n) =

n/2 nが偶数のとき

3n+ 1 nが奇数のとき

このときCollatzの予想と呼ばれるものは，f(n), f2(n), f3(n), . . . と繰

り返していけば，どのような正の整数 nに対しても必ず 1に到達すると

いうものである．この予想は 3× 253 = 27, 021, 597, 764, 222, 976以下

で正しいことが知られている．それでは，この予想が正しいとして，1

より大きな整数 nが与えられたとき，fk(n) = 1となる最小の k > 1を

求める関数 nankaiを作れ．また，nが 2から 100までの，nankai(n)

の値のリストを作れ．

